
HSL ZB01
PACKAGE SPECIFICATION HSL

1 SUMMARY

Given a rank-one or rank-two allocatable array, HSL ZB01 reallocates the array to have a different size, and can copy
all or part of the original array into the new array. This will use a temporary array or, if there is insufficient memory,
one or more temporary or scratch files will be used. The user may force HSL ZB01 to only use files and not to attempt
to use a temporary array. The user may optionally supply the name of the temporary file and the file size; if no name
is supplied, a scratch file is used. If more than one file is required and a filename has been supplied, then HSL ZB01
opens files with names that it constructs from the filename by appending ‘1’, ‘2’,. . . to the end. Any temporary files
are deleted upon successful exit.

ATTRIBUTES — Version: 3.0.0. Types: Real (single, double), Complex (single, double), Integer (default, long).
Calls: None. Language: Fortran 95 + TR 15581 (allocatable components). Date: Original date July 2007. Version
2.0.0 December 2010 Origin: H. S. Dollar, Rutherford Appleton Laboratory. Remark: The development of this
package was supported by EPSRC grant GR/S42170.

2 HOW TO USE THE PACKAGE

2.1 Calling sequences

Access to the package requires a USE statement of the form

Single precision version
USE HSL ZB01 single

Double precision version
USE HSL ZB01 double

Default integer version
USE HSL ZB01 integer

Long integer version
USE HSL ZB01 long integer

Complex version
USE HSL ZB01 complex

Double complex version
USE HSL ZB01 double complex

If it is required to use two or more modules at the same time, the derived type ZB01 info (Section 2.3.3) and the
subroutines ZB01 resize1 and ZB01 resize2 must be renamed in one of the USE statements.
The following procedures are available to the user:

• ZB01 resize1 resizes a rank-one array.

• ZB01 resize2 resizes a rank-two array.

In the following the package type denotes
Default REAL in HSL ZB01 single

REAL(kind = kind(0.0d0)) in HSL ZB01 double

Default INTEGER in HSL ZB01 integer

INTEGER(kind = selected int kind(18)) in HSL ZB01 long integer

Default COMPLEX in HSL ZB01 complex

COMPLEX(kind = kind(0.0d0)) in HSL ZB01 double complex

When describing the types of the arguments, INTEGER(long) denotes INTEGER(kind = selected int kind(18)).

2.2 The derived data types

For each problem, the user must employ the derived type defined by the module to declare a scalar of the type
ZB01 info. The following pseudocode illustrates this.

use HSL_ZB01_double
...
type (ZB01_info) :: info
...

The user may optionally employ a derived type defined by the module to declare a scalar of the type ZB01 control.
The following pseudocode illustrates this.

use HSL_ZB01_double
...
type (ZB01_control) :: control
...

The components of ZB01 info and ZB01 control are explained in Section 2.3.3 and Section 2.3.4, respectively.

2.3 The argument lists

We use square brackets [ ] to indicate OPTIONAL arguments. In each call, optional arguments follow the argument
info. Since we reserve the right to add additional optional arguments in future releases of the code, we strongly
recommend that all optional arguments be called by keyword, not by position.

2.3.1 To resize a rank-one array

To reallocate a given rank-one to have a different size, and optionally copy all or part of the original array into the new
array, a call of the following form may be made

CALL ZB01_resize1(w,size_in,size_out,info[,src,dest,filename,file_size,mode,control])

w is a rank-one allocatable array of INTENT(INOUT) and package type to be reallocated. On entry, w must be allocated
to have size size in. On exit, w will have size size out. If src and/or dest are present, the specified entries
are copied into the reallocated array; if neither both src nor dest are present, the original entries are preserved.

size in is a scalar INTENT(IN) argument of type INTEGER(long). It specifies the initial size of w. Restriction:
size in≥1.

size out is a scalar INTENT(INOUT) argument of type INTEGER(long). On entry, it specifies the requested size of w.
On exit, size out specifies the size of w after reallocation. If the routine cannot allocate w to have the requested
size, it will attempt to allocate a smaller array that contains the entries to be copied. Restriction: if dest is
absent, size out≥size in.

info is a scalar INTENT(OUT) argument of type ZB01 info. info%flag is used as an error/warning flag. Negative
values indicate an error and positive values a warning. For details of the components of info see Section 2.3.3.

src and dest are rank-one OPTIONAL array arguments of INTENT(IN), type INTEGER(long) and size 2. If src and
dest are present, w(src(1):src(2)) are copied to w(dest(1):dest(2)). The array sections described by
src and dest must have the same size, but may overlap. If only src or dest is present, it is equivalent to passing
both arguments with the same value. If neither is present, all entries are preserved. If src(2)<src(1) then no
copying will occur. Restriction: minval(src)>0, minval(dest)>0, dest(2)-dest(1)=src(2)-src(1); if
src(2)≥src(1), then src(2)≤size in; if dest(2)≥dest(1), then dest(2)≤size out.

filename is an OPTIONAL scalar INTENT(IN) argument and type default CHARACTER with length at most 400. If
filename is absent, a scratch file will be used. On entry, filename must contain the path and name to be
used when creating temporary files. If more than one file is required and filename is present, HSL ZB01
opens files with names that it constructs from filename by appending ‘1’, ‘2’,. . . to the end. Restriction:
len(filename)≤400, files with names of this form must not already exist.

file size is an OPTIONAL scalar of INTENT(IN) and type INTEGER(LONG). If present, file size must be set to the
target size of each file, measured in scalars of the package type. If absent, the value 222 is used. Restriction:
file size≥ 212. The upper bound on file size is machine and compiler dependent.

mode is an OPTIONAL scalar of INTENT(IN) and type INTEGER. If mode=0, a temporary array will be used or, if there
is insufficient memory, temporary or scratch files will be used. If mode=1, only temporary or scratch files will
be used. If mode is absent, then the behaviour is as if it were present with value 0. Restriction: mode=0 or 1.

control is an OPTIONAL scalar argument of type ZB01 control, which controls the behaviour of the program. For
details of the components of control see Section 2.3.4.

2.3.2 To resize a rank-two array

To reallocate a given rank-two array to have a different size, and optionally copy all or part of the original array into
the new array, a call of the following form may be made

CALL ZB01_resize2(w,size_in,size_out,info[,src,dest,filename,file_size,mode,control])

w is a rank-two allocatable array of INTENT(INOUT) and package type to be reallocated. On entry, w must be allocate
to have shape (size in(1),size in(2)). On exit, w will have shape (size out(1),size out(2)). If src
and/or dest are present, the specified entries are copied into the reallocated array; if neither src nor dest are
present, the original entries are unchanged.

size in is a rank-one array INTENT(IN) argument of type INTEGER(long) and size (2). It specifies the initial shape
of w. Restriction: size in≥1.

size out is a rank-one array argument of INTENT(INOUT), type INTEGER(long) and size (2). On entry, it specifies
the requested shape of w. On exit, size out specifies the shape of w after reallocation. If the routine cannot
allocate w to have the requested shape, it will attempt to allocate a smaller array that contains the entries to be
copied. Restriction: If dest is absent, size out(1)≥size in(1) and size out≥size in(2).

info is as described in Section 2.3.1.

src and dest are OPTIONAL arrays of INTENT(IN) and type INTEGER(long) that both have size (2,2). If both are
present, the entries w(src(1,1):src(2,1), src(1,2):src(2,2)) are copied to w(dest(1,1):dest(2,1),
dest(1,2):dest(2,2)). The array sections described by src and dest must have the same size, but may
overlap. If only src or dest is present, it is equivalent to passing both arguments with the same value. If neither
is present, all entries are preserved. If src(2,1)≤src(2,2) or dest(2,1)≤dest(2,2) then no copying will
occur. Restriction: minval(src)>0, minval(dest)>0, dest(2,:)-dest(1,:)=src(2,:)-src(1,:); if
all(src(2,:)≥src(1,:)), then src(2,:)≤size in(:); if all(dest(2,:)≥dest(1,:)), then
dest(2,:)≤size out(:).

filename is as described in Section 2.3.1.

file size is as described in Section 2.3.1.

mode is as described in Section 2.3.1.

control is as described in Section 2.3.1.

2.3.3 The derived data type for holding information

The derived data type ZB01 info is used to hold information from the execution of ZB01. The components are:

flag is an INTEGER scalar that is used as an error/warning flag. Negative values indicate an error. Possible negative
values for flag are:

-1 Destination for copied data exceeds upper bound of new array
In ZB01 resize1: 1≤size out<maxval(dest) if dest is present and dest(2)≥dest(1); or
1≤size out<size in if dest is absent.
In ZB01 resize2: 1≤size out(1)<maxval(dest(:,1)) and 1≤size out(2)<maxval(dest(:,2))
if dest is present and dest(2,1)≥dest(1,1) and dest(2,2)≥dest(1,2); or 1≤size out(1)<size in(1)
and 1≤size out(2)<size in(2) if dest is absent.

-2 In ZB01 resize1: size out<1.
In ZB01 resize2: size out(1)<1 or size out(2)<1.

-3 Combination of -1 and -2.

-4 Source for copied data exceeds upper bound of original array
In ZB01 resize1: maxval(src)>size in
In ZB01 resize2: maxval(src(:,1))>size in(1) or maxval(src(:,2))>size in(2).

-5 minval(src)<0 or minval(dest)<0.

-6 Combination of -4 and -5.

-7 len(filename)>400.

-8 file size< 212.

-9 file of the form filename\\’1’ already exists.

-10 mode<0 or mode>1.

-11 unrecoverable memory allocation failure. The stat parameter is returned in info%stat.

-12 memory deallocation failure. The stat parameter is returned in info%stat.

-13 error in Fortran inquire statement. The Fortran iostat parameter is returned in info%iostat.

-14 error in Fortran open statement. The Fortran iostat parameter is returned in info%iostat.

-15 error in Fortran read statement. The Fortran iostat parameter is returned in info%iostat.

-16 error in Fortran write statement. The Fortran iostat parameter is returned in info%iostat.
-17 error in Fortran close statement. The Fortran iostat parameter is returned in info%iostat.
-18 the array sections described by src and dest do not have compatible sizes.
-19 The Array w was unallocated on entry.

Positive values indicate an warning. Possible positive values for flag are:

+1 size out was changed by the routine.
+2 w was not allocated on input.

iostat is a scalar of type INTEGER that holds the Fortran iostat parameter.

stat is a scalar of type INTEGER that holds the Fortran stat parameter.

files used is a scalar of type INTEGER that holds the number of temporary files that were used during the execution
of ZB01 resize1 or ZB01 resize2.

2.3.4 The derived data type for holding control parameters

The derived data type ZB01 control is used to hold parameters used to control the behaviour of ZB01. The compo-
nents are:

print level is an INTEGER scalar that controls whether or not diagnostic printing should be enabled. Positive values
print messages about the status of the software, while printing is supressed for all other values. The default
value is print level = 0.

print unit is an INTEGER scalar that holds the unit number for diagnostic printing. The default value is print unit
= 6.

test no mem w is a scalar of type LOGICAL allows the user to test whether the initial reallocation of w has failed.
This is for diagnostic and testing use only, and should not be set to .true. in production. The default value is
test no mem w = .false..

test still no mem w is a scalar of type LOGICAL allows the user to test whether the second reallocation of w has
failed. This is for diagnostic and testing use only, and should not be set to .true. in production. The default
value is test still no mem w = .false..

test more no mem w is a scalar of type LOGICAL allows the user to test whether the third reallocation of w has failed.
This is for diagnostic and testing use only, and should not be set to .true. in production. The default value is
test more no mem w = .false..

3 GENERAL INFORMATION

Input/output: I/O to sequential-access files whose unit numbers are chosen by HSL ZB01.

Restrictions: In both ZB01 resize1 and ZB01 resize2: size out≥1, len(filename)≤400, file size≥ 212,
mode=0 or 1.

In ZB01 resize1: minval(src)>0, minval(dest)>0, dest(2)-dest(1)=src(2)-src(1),
if dest(2)≥dest(1) then dest(2)≤size out, if src(2)≥src(1) then src(2)≤size in.

In ZB01 resize2: minval(src)>0, minval(dest)>0, dest(2,:)-dest(1,:)=src(2,:)-src(1,:),
if all(dest(2,:)≥dest(1,:) then dest(2,:)≤size out(:), if all(src(2,:)≥src(1,:) then
src(2,:)≤size in(:),

Changes from Version 1 Version 1 used default integers for array sizes and only permitted preservation of the first
lkeep entries of w; Version 2 changed to long integers for array sizes and the original lkeep argument was
replaced by the src and dest arguments. Version 2 also removed the ability to resize an unallocated array
present in Version 1.

4 METHOD

This package changes the size of a given array w and can copy all or part of the original array into the new array.
If no entries are to be copied, then the original array is deallocated: if this fails, then info%flag=-12 is returned.

The array is then reallocated: if this fails, then info%flag=-11 is returned.
Otherwise, if mode is absent or mode=0, then an attempt is made to copy these entries into a temporary array. If

allocation of this temporary array fails, or mode=1, the entries are copied into a sequential access file. If filename is
absent, this file has status=’scratch’. More than one file may be used. The user may optionally set the target size
of the file. Small values of file size may result in a large number of files being written. The original array is then
deallocated: if this fails, info%flag=-12 is returned. The array is then reallocated: if this fails and a temporary array
was used, the entries in the temporary array are copied into temporary files and the temporary array deallocated before
attempting to reallocate the original array again. If allocation is still unsuccessful, then the array is reallocated to have
size/shape equal to the size/shape of the portion of the original array being copied. If this is successful, then size out
is updated and info%flag=+1 is returned; otherwise info%flag=-11 is returned. Finally, the desired entries are
copied back into this array in the appropriate positions and the temporary array/file is deallocated/deleted.

The method may fail if file size is too large or if the space available on the device to which files are being
written is insufficient. If info%flag>-13, temporary files have are deleted on exit.

5 EXAMPLE OF USE

We illustrate the use of the package on both a rank-one and a rank-two array.

Example 1
In this example a rank-one array is initialised to have size 5 and entries 1,2,3,4,5. The array is then reallocated to have
size 10 and the first 3 entries from the original array are copied into the new array.

program example1
use hsl_zb01_double
implicit none

integer, parameter :: wp = kind(0d0)
integer, parameter :: long = selected_int_kind(18)

integer(long) :: i, size_in, size_out, src(2)
real(wp), allocatable, dimension(:) :: w
type(zb01_info) :: info

! Allocate original array and initialise entries.
size_in = 5
allocate (w(size_in))
do i = 1, size_in

w(i) = i
end do

! Resize array to have length 10 and keep first 3 entries.
size_out = 10
src(:) = (/1,3/)
call zb01_resize1(w,size_in,size_out,info,src=src)

write (*,’(a)’) ’Size of array:’
write (*,’(i10)’) size(w)

write (*,’(a)’) ’Number of temporary files used:’
write (*,’(i10)’) info%files_used

write (*,’(a)’) ’Contents of first 3 entries of array:’
write (*,’(5f10.1)’) w(1:3)

end program example1

Output

Size of array:
10

Number of temporary files used:
0

Contents of first 3 entries of array:
1.0 2.0 3.0

Example 2
In this second simple example, a rank-two array w is initialised to have 3 rows and 3 columns and entries w(i,j) =
10×i + j. The array is then reallocated to have 3 rows and 5 columns, and the entries from first two rows of the
original array are copied into the new array. We request that temporary files are used without attempting to allocate a
temporary array (mode=1).

program example2
use hsl_zb01_double
implicit none

integer, parameter :: wp = kind(0d0)
integer, parameter :: long = selected_int_kind(18)

integer(long) :: i, j, size_in(2), size_out(2), src(2,2)
real(wp), allocatable, dimension(:,:) :: w
type(zb01_info) :: info

! Allocate original array and initialise entries.
size_in = (/ 3_long, 3_long /)
allocate (w(size_in(1),size_in(2)))
do i = 1, size_in(2)

do j = 1, size_in(1)
w(i,j) = 10*i + j

end do
end do

! Resize array to have 3 rows, 5 columns and keep entries in

! intersection of first 2 rows and 3 columns.
! Use temporary scratch files.
size_out(:) = (/ 3_long, 5_long /)
src(:,1) = (/ 1_long, 2_long /); src(:,2) = (/ 1_long, size_in(2) /)
call zb01_resize2(w,size_in,size_out,info,src=src,mode=1)

write (*,’(a)’) ’Shape of array:’
write (*,’(2i10)’) size(w,1), size(w,2)

write (*,’(a)’) ’Number of temporary files used:’
write (*,’(i10)’) info%files_used

write (*,’(a)’) ’Contents of first 2 rows and 3 columns of array:’
do i = 1, 2

write (*,’(5f10.1)’) w(i,1:3)
end do

end program example2

Output

Shape of array:
3 5

Number of temporary files used:
1

Contents of first 2 rows and 3 columns of array:
11.0 12.0 13.0
21.0 22.0 23.0

